
In diesem Beispiel werden Folien aus der PPT-Anwendung heraus in eine 

Worddokument exportiert und als PNG-Dateien in einen Ordner gespeichert. 

1 PPT-Folien in ein Worddokument exportieren 

Die folgende Prozedur überträgt die Folien einer PPT-Präsentation über die 

Zwischenablage in ein Worddokument. Im VBA-Editor der PPT-Präsentation wird unter 

Extras / Verweise die Microsoft Word 16.0 Object Library für eine frühe Bindung 

ausgewählt. Ein eingefügtes Modul erhält die folgende Prozedur. 

Codeliste 1. Export der Folien als Bilddateien in ein Worddokument 

Sub FolienToWord() 

   Dim objSld As PowerPoint.Slide 

   Dim objWrd As Word.Application 

   Dim objDoc As Word.Document 

   Dim rngTmp As Word.Range 

    

   Set objWrd = New Word.Application 

   objWrd.Visible = True 

   Set objDoc = objWrd.Documents.Add 

   objDoc.Activate 

   Set rngTmp = objDoc.Range 

   For Each objSld In ActivePresentation.Slides 

      objSld.Copy 

      rngTmp.PasteSpecial datatype:=4 '14 umgekehrt 

   Next 

   Set objDoc = Nothing 

   Set objWrd = Nothing 

End Sub 

Nach der Ausführung befinden sich die Bildelemente in Folge im Dokument (Bild 1). 

 

Bild 1. Darstellung der Folien als Bildelemente 

Zur Darstellung in umgekehrter Reihenfolge muss Datatype:=14 angegeben werden. 

2 Folien als Bilder in einen Ordner exportieren 

Mit der Export-Methode des Slide-Objekts können Folien als Bilddateien im PNG-, 

JPG- oder BMP-Format erstellt werden. Außerdem kann beim Export das Bildformat 

verändert werden. Der Defaultwert von 1280 x 720 Pixel kann durch Faktoren wie 

0.5 (= 50%) umgestellt werden.  

Codeliste 2. Export der Folien als Bilddateien in einem vorgegebenen Ordner 

Sub FolienExport() 

   Dim objSld  As PowerPoint.Slide 

   Dim sTyp    As String 

   Dim sNam    As String 

   Dim sOrd    As String 

   Dim iFol    As Integer 

   Dim iWidth  As Integer 

   Dim iHeight As Integer 

    

   sOrd = "C:\Temp\Folie" 

   sTyp = "png" ' oder jpg, bmp ... 

   iWidth = 1280 * 0.5 

   iHeight = 720 * 0.5 

   For iFol = 1 To ActivePresentation.Slides.Count 

      Set objSld = ActivePresentation.Slides(iFol) 

      sNam = sOrd & Trim(Str(iFol)) & "." & sTyp 

      objSld.Export sNam, sTyp, _ 

         ScaleWidth:=iWidth, ScaleHeight:=iHeight 

   Next 

End Sub 

Nach der Ausführung befinden sich die Bilddateien in dem vorgegebenen Ordner 

(Bild 2). 

 

Bild 2. Exportierte Folien 


